

Challenge November 2025

Advanced Website Design

A solution with OPL CPLEX by Alex Fleischer
afleischer@fr.ibm.com

CPLEX is available within IBM AI platform watsonx.AI
And OPL is a high level modeling language for CPLEX

We add the additional constraints to the October 2025 OPL CPLEX

```
tuple feature
{
    key int id;
    int cost;
    int value;
}

{feature} features=
{
    <1,5000,7>,
    <2,4000,6>,
    <3,3000,5>,
    <4,2000,4>,
    <5,1000,3>,
    <6,1000,2>,
    <7,1000,1>,
    <8,1000,1>,
    <9,1000,1>,
    <10,1000,1>
};

float budget=10000;

dvar boolean x[features];

dvar float obj1;
dvar float obj2;
dvar float cost0;
dvar float cost1;
dvar float cost2;
dvar boolean onlyoneofthefeatures8910;

// not dominated
maximize obj1+obj2;
subject to
{

    // obj1 is the number of features
    obj1==sum(f in features) x[f];
    // obj2 is the total value of the features
```

```

    obj2==sum(f in features) x[f]*f.value;

// In the November challenge we add additional constraints

//x[<1>]==0;
//x[<2>]==0;
//x[<8>]==0;
//x[<9>]==0;


//    Features 3 and 4 can be chosen only together.

x[<3>]==x[<4>];

//    Feature 2 cannot be combined with Feature 3.

x[<2>]+x[<3>]<=1;

//    Only one of the features 8, 9, and 10 can be selected.
//    Otherwise, the cost of each of these features will be increased by 10%.

onlyoneofthefeatures8910==(x[<8>]+x[<9>]+x[<10>]<=1);

//    If 5 or more features are selected, 5% discount is provided.

cost0==sum(f in features) x[f]*f.cost;
(onlyoneofthefeatures8910==1) =>
(cost1==cost0);
(onlyoneofthefeatures8910==0) =>
(cost1==cost0+0.1*sum(f in features:f.id in 8..10)x[f]*f.cost);

(obj1>=5) => (cost2==cost1*0.95);
(obj1<=4) => (cost2==cost1);

// cost should be less than budget
cost2<=budget;
}

execute display
{
    writeln("obj1=",obj1);
    writeln("obj2=",obj2);
    writeln("features : ");
    for(var f in features) if (x[f]==1) write(f.id," ");
    writeln();
    writeln("cost0=",cost0);
    writeln("cost1=",cost1);
    writeln("cost2=",cost2);
}

main
{

```

```

//writeln("solution pools")

thisOplModel.generate();
cplex.solnpoolintensity=4;
cplex.solnpoolcapacity=1000;
cplex.solnpoolreplace=2;
cplex.populatelim=2000;

cplex.populate();
var nsolns = cplex.solnPoolNsolns;
//writeln("n sol ",nsolns);

var obj1=new Array(nsolns);
var obj2=new Array(nsolns);;

for(var s = 0; s < nsolns; s++)
{
    thisOplModel.setPoolSolution(s);
    //thisOplModel.postProcess();
    obj1[s]=thisOplModel.obj1.solutionValue;
    obj2[s]=thisOplModel.obj2.solutionValue;

}

writeln("not dominated solutions");
writeln();

for(var s = 0; s < nsolns; s++)
    for(var s2 = s+1; s2 < nsolns; s2++) if ((obj1[s]==obj1[s2]) &&
(obj2[s]==obj2[s2]))
    {
        obj1[s2]=0;
        obj2[s2]=0;
    }

for(var s = 0; s < nsolns; s++)
{
    var dominated=0;
    for(var s2 = 0; s2 < nsolns; s2++) if (s!=s2) if ((obj1[s2]>=obj1[s]) &&
(obj2[s2]>=obj2[s]))
        dominated=1;
    if (dominated==0)
    {
        thisOplModel.setPoolSolution(s);
        thisOplModel.postProcess();
        writeln();
    }
}
cplex.clearModel();
}

```

Which gives

not dominated solutions

obj1=7
obj2=17
features :
3 4 5 6 8 9 10
cost0=10000
cost1=10300
cost2=9785